



Vaults Solver Midas Audit

Côme du Crest

2025-10-12

Table of contents

- Table of contents
- Vaults Solver Midas Audit
 - Scope
 - Context
 - Status
 - Legal Information And Disclaimer
- Issues
 - [Fixed][High] instantRedeem does not charge a fee
 - [Fixed][High] zapVaults transfers incorrect fee
 - [Fixed][Low] Admin fee setting documented in percent, used in bps
 - [Fixed][Low] Conversion using Midas oracles may be incorrect for stable coins
 - [Ack][Info] Midas fee is ignored
 - [Ack][Info] Complete reliance on Midas oracles
 - Optimisations and miscellaneous

Vaults Solver Midas Audit

This document presents the findings of a smart contract audit conducted by Côme du Crest.

For any inquiry contact me by email at come.ducrest@gmail.com or via telegram [@ComeduCrest](https://t.me/ComeduCrest).

Scope

The scope includes the single `VaultsSolverMidas` contract within `0xhyperbeat/DNCoreWriter` as of commit `0xb0e409d`.

Fixes have been implemented and the contract has been reviewed as of commit `0xb8059fc`

Context

The goal of `VaultsSolverMidas` is to allow users to instantly redeem mTokens from Midas into a different supported withdrawal tokens. This is done through the `instantRedeem()` function that converts the input mToken amount to USD and the USD value to output token amount before transferring the tokens to the recipient.

The contract also has a `deposit()` function to deposit accepted tokens in exchange for mTokens. The conversion is similar to the `instantRedeem()` function in that it converts the deposited token amount to USD value and the USD value to mToken output amount.

The last user function is `zapVaults()` which allows to deposit an mToken and receive a different mToken with the same conversion logic as the two previous functions.

Both `instantRedeem()` and `zapVaults()` charge a fee while `deposit()` does not.

The intent is that the mTokens and other output tokens are deposited by the admin of the contract.

The oracles for converting tokens values to and from USD are the ones used by the underlying Midas vaults.

Status

The report has been sent to the core developer.

All issues have been fixed or acknowledged and no issue remain.

Legal Information And Disclaimer

1. This report is based solely on the information provided by Dunamis Labs Ltd (the “Company”), with the assumption that the information provided is authentic, accurate, complete, and not misleading as of the date of this report. The auditor has not conducted any independent enquiries, investigations or due diligence in respect of the Company, its business or its operations.
2. Changes to the information contained in the documents, repositories and any other materials referenced in this report might affect or change the analysis and conclusions presented. The auditor is not responsible for monitoring, nor will we be aware of, any future additions, modifications, or deletions to the audited code. As such, the auditor does not assume any responsibility to update any information, content or data contained in this report following the date of its publication.
3. This report does not address, nor should it be interpreted as addressing, any regulatory, tax or legal matters, including but not limited to: tax treatment, tax consequences, levies, duties, data privacy, data protection laws, issues relating to the licensing of information technology, intellectual property, money laundering and countering the financing of terrorism, or any other legal restrictions or prohibitions. The auditor disclaims any liability for omissions or errors in the findings or conclusions presented in this report.
4. The views expressed in this report are solely my views regarding the specific issues discussed within this report. This report is not intended to be exhaustive, nor should it be construed as an assurance, guarantee or warranty that the code is free from bugs, vulnerabilities, defects or deficiencies. Different use cases may carry different risks, and integration with third-party applications may introduce additional risks.
5. This report is provided for informational purposes only and should not be used as the basis for making investment or financial decisions. This report does not constitute investment research and should not be viewed as an invitation, recommendation, solicitation or offer to subscribe for or purchase any securities, investments, products or services. The auditor is not a financial advisor, and this report does not constitute financial or investment advice.
6. The statements in this report should be considered as a whole, and no individual statement should be extracted or referenced independently.
7. To the fullest extent permitted by applicable laws, the auditor disclaims any and all other liability, whether in contract, tort, or otherwise, that may arise from this report or the use thereof.

Issues

[Fixed][High] `instantRedeem()` does not charge a fee

Summary

The function `instantRedeem()` does not lower the amount of tokens sent to the caller by the computed fee value. As such, it does not actually charge a fee to the user.

Vulnerability Detail

The function `instantRedeem()` is supposed to charge a fee to the user. It computes the fee value `instantRedeemFee` from the `tokenAmount` amount of tokens corresponding to the user's given amount of mTokens. However it transfers the total `tokenAmount` assets to the user and did not take out a fee:

```
1     function instantRedeem(  
2         address _recipient,  
3         address _vaultToken,  
4         address _withdrawalAsset,  
5         uint256 _amount  
6     )  
7     external  
8     {  
9         if (!supportedWithdrawalAssetsPerVault[_vaultToken][_withdrawalAsset])  
10            {  
11                revert VaultInstantWithdrawalMidas__WithdrawalAssetNotSupported(  
12                    _vaultToken, _withdrawalAsset);  
13            }  
14            (uint256 usdAmount, address redemptionVault) = _convertMTokenToUSD(  
15                _vaultToken, _amount);  
16            uint256 tokenAmount = _convertUSDToToken(redemptionVault,  
17                _withdrawalAsset, usdAmount);  
18            uint256 instantRedeemFee = _calcInstantRedeemFee(_vaultToken,  
19                tokenAmount);  
20            uint256 adminFee = _calcAdminFee(instantRedeemFee);  
21            if (adminFee > 0 && adminFeeRecipient != address(0)) {  
22                _withdrawalAsset.safeTransfer(adminFeeRecipient, adminFee);  
23            }  
24            uint256 amountToTransfer = _amount - instantRedeemFee;  
25            _vaultToken.safeTransferFrom(msg.sender, address(this), _amount);  
26            _withdrawalAsset.safeTransfer(_recipient, tokenAmount); // @audit no  
27                fee taken  
28            emit InstantRedeem(_vaultToken, _withdrawalAsset, _amount,  
29                amountToTransfer, instantRedeemFee, adminFee);  
30        }
```

Note that the value `amountToTransfer` is strictly used in the event and does not represent any-

thing as the `_amount` it is computed from is in `mToken` and `instantRedeemFee` is in output asset value.

Impact

No fee is actually charged to the user and the protocol realises a loss in this conversion as Midas actually charges a fee for such actions.

Code Snippets

<https://github.com/0xhyperbeat/DNCoreWriter/blob/b0e409d6e53ebf17cf881f3a323acdcd60c47b85/src/VaultsSolverMidas.sol#L87-L109>

<https://github.com/0xhyperbeat/DNCoreWriter/blob/b0e409d6e53ebf17cf881f3a323acdcd60c47b85/tests/VaultsSolverMidas.t.sol#L150>

Recommendation

The amount of withdrawal asset that should be sent to the user should be `tokenAmount - instantRedeemFee`.

Fix the tests by recomputing the amount expected to be received by the user instead of using a broad `assertGt()` to assert the received any amount of tokens. You can use `assertApproxEqAbs()` as an assert method (see [docs](#)).

Response

This has been fixed as of commit [0x3f6115a](#) by taking the fee in `mToken` and computing the USD value based on `_amount - instantRedeemFee`.

[Fixed][High] zapVaults transfers incorrect fee

Summary

The function `zapVaults()` compute a fee based on the input amount of tokens to swap but transfers the output tokens to the fee recipient instead.

Vulnerability Detail

In `zapVaults()`, a fee is computed based on the amount of input token. The `amountAfterFee` value is computed from the input amount token minus the `zapFee`. This amount after fee is used as the base for the calculation of the amount of tokens to output.

However, the admin fee tokens are transferred to the admin fee recipient using `_toVaultToken.safeTransfer()` meaning that the fees are taken in the wrong token:

```
1      function zapVaults(address _fromVaultToken, address _toVaultToken, uint256
      _amount) external {
2          ...
3          uint256 zapFee = _calcZapFee(_fromVaultToken, _toVaultToken, _amount);
4          uint256 adminFee = _calcAdminFee(zapFee);
5          if (adminFee > 0 && adminFeeRecipient != address(0)) {
6              _toVaultToken.safeTransfer(adminFeeRecipient, adminFee);
7          }
8          uint256 amountAfterFee = _amount - zapFee;
9          (uint256 fromUSDAmount,) = _convertMTokenToUSD(_fromVaultToken,
              amountAfterFee);
10         address depositVault = depositVaults[_toVaultToken];
11         if (depositVault == address(0)) {
12             revert VaultInstantWithdrawalMidas__DepositVaultNotSet(
                _toVaultToken);
13         }
14         uint256 toMTokenAmount = _convertUSDToMToken(depositVault,
              fromUSDAmount);
15         _fromVaultToken.safeTransferFrom(msg.sender, address(this), _amount);
16         _toVaultToken.safeTransfer(msg.sender, toMTokenAmount);
17         emit ZapVaults(_fromVaultToken, _toVaultToken, _amount, toMTokenAmount,
              zapFee, adminFee);
18     }
```

Impact

If 10 mTokenA converts into 1.000 mTokenB and the admin fee percent is set to 10%, a fee of 1 mTokenA should be sent to the admin fee recipient. However, a fee of 1 mTokenB will be sent instead.

The fee sent to the admin fee recipient uses the incorrect token.

Code Snippets

<https://github.com/0xhyperbeat/DNCoreWriter/blob/b0e409d6e53ebf17cf881f3a323acdcd60c47b85/src/VaultsSolverMidas.sol#L140>

Recommendation

Transfer the correct token to the admin fee recipient (`_fromVaultToken`).

Response

This has been fixed as of commit `0x3f6115a` by transferring `_fromVaultToken` as admin fee.

[Fixed][Low] Admin fee setting documented in percent, used in bps

Summary

The admin fee setting is documented to be provided in percent but is actually used in bps. The documentation of `_calcAdminFee()`, `setAdminFee()`, and the variable name itself `adminFeePct` indicate using a percent calculation. However the `adminFeePct` value is divided by `1e4` which is a bps calculation.

Vulnerability Detail

The documentation states using a percent fee value but the calculation uses a bps value:

```
1  contract VaultsSolverMidas ... {
2      ...
3      uint256 public constant BASE = 1e4;
4      uint256 public adminFeePct;
5
6      /// @notice Calculates the admin fee from a given amount
7      /// @dev Fee is calculated as a percentage of the amount using the global
8      /// admin fee rate
9      /// @param _amount Amount to calculate fee on
10     /// @return fee The calculated admin fee
11     function _calcAdminFee(uint256 _amount) internal view returns (uint256 fee)
12     {
13         fee = _amount.mulDiv(adminFeePct, BASE);
14     }
15
16     /// @notice Sets the global admin fee percentage
17     /// @dev Only callable by ADMIN_ROLE. Fee is expressed in basis points
18     /// relative to BASE (1e4)
19     /// @param _adminFee Fee percentage in basis points
20     function setAdminFee(uint256 _adminFee) external onlyRole(ADMIN_ROLE) {
21         adminFeePct = _adminFee;
22         emit AdminFeeSet(_adminFee);
23     }
24 }
```

Impact

If the admin sets the fee expecting a percent value to be used they will receive less fee than expected.

Code Snippets

<https://github.com/0xhyperbeat/DNCoreWriter/blob/b0e409d6e53ebf17cf881f3a323acdcd60c47b85/src/VaultsSolverMidas.sol#L253-L255>

Recommendation

Make the documentation explicit about expecting a bps value instead of a percent value.

Response

The variable `adminFeePct` has been renamed `adminFeeBps` as of commit `0x3f6115a`. The notice of the natspec of `setAdminFee()` and `_calcAdminFee()` still refer to percents.

[Fixed][Low] Conversion using Midas oracles may be incorrect for stable coins

Summary

The `VaultsSolverMidas` attempts to mimic the conversion rates of Midas vaults by using the `getDataInBase18()` function from its data feed. However, Midas vaults ignore the value received from these feeds if the converted token is a stable coin and use a constant `STABLECOIN_RATE = 1e18` value instead.

If Midas vaults are set with data feeds for stable coins that do not return `1e18` there will be a discrepancy in between the conversion rate used by `VaultsSolverMidas` and the one used by Midas vaults.

Vulnerability Detail

The function `ManageableVault._getTokenRate()` used by Midas vaults to get the exchange rate of a token overwrites the answer from the data feed by `1e18` if the converted token is a stable coin:

```
1      uint256 public constant STABLECOIN_RATE = 10**18;
2
3      function _getTokenRate(address dataFeed, bool stable)
4          internal
5          view
6          virtual
7          returns (uint256)
8      {
9          // @dev if dataFeed returns rate, all peg checks passed
10         uint256 rate = IDataFeed(dataFeed).getDataInBase18();
11
12         if (stable) return STABLECOIN_RATE;
13
14         return rate;
15     }
```

This functions is used for example to convert the value of a token to USD in `DepositVault`:

```
1      function _convertTokenToUsd(address tokenIn, uint256 amount)
2          internal
3          view
4          virtual
5          returns (uint256 amountInUsd, uint256 rate)
6      {
7          require(amount > 0, "DV: amount zero");
8
9          TokenConfig storage tokenConfig = tokensConfig[tokenIn];
10
11         rate = _getTokenRate(tokenConfig.dataFeed, tokenConfig.stable);
12         require(rate > 0, "DV: rate zero");
13     }
```

```
14     amountInUsd = (amount * rate) / (10**18);
15 }
```

However, `VaultsSolverMidas._convertTokenToUSD()` directly uses the underlying `getDataInBase18()` without considering whether the token is a stable coin or not:

```
1  function _convertTokenToUSD(
2      address _vaultToken,
3      address _depositAsset,
4      uint256 _amount
5  )
6      internal
7      view
8      returns (uint256 usdAmount, address depositVault)
9  {
10     depositVault = depositVaults[_vaultToken];
11     if (depositVault == address(0)) {
12         revert VaultInstantWithdrawalMidas__DepositVaultNotSet(_vaultToken)
13         ;
14     }
15     IMidasDepositVault.TokenConfig memory tokenConfig = IMidasDepositVault(
16         depositVault).tokensConfig(_depositAsset);
17     usdAmount = _amount.mulDiv(IMidasDataFeed(tokenConfig.dataFeed).
18         getDataInBase18(), 10 ** 18);
19 }
```

Impact

If `VaultsSolverMidas` is used with stable coins, the conversion rate used may be different from what is used in the underlying Midas vaults.

Users can abuse this discrepancy for example by getting cheap mTokens for overpriced stable coins on `VaultsSolverMidas` and withdrawing stable coins on Midas vaults before repeating the operation.

Code Snippets

<https://github.com/0xhyperbeat/DNCoreWriter/blob/b0e409d6e53ebf17cf881f3a323acdcd60c47b85/src/VaultsSolverMidas.sol#L223-L238>

<https://github.com/midas-apps/contracts/blob/ba8596751be53f4461349316b380c1e7868782e5/contracts/abstract/ManageableVault.sol#L632-L644>

Recommendation

Use the value `tokenConfig.stable` from the config received from the underlying Midas vault to determine whether the data feed's rate should be overwritten with `1e18`.

Response

Fixed as of commit `0xb8059fc` by overwriting the Midas oracles responses with `1e18` if the coin is configured as stable.

[Ack][Info] Midas fee is ignored

Summary

The `deposit()` function that allows to deposit an asset and receive mTokens in exchange does not charge a fee. It is more beneficial for users to deposit through `VaultsSolverMidas` than to go through the official Midas `DepositVault` deposit process which charges its own fee.

This means that if the `VaultsSolverMidas` intends to convert the deposited assets into mToken through Midas, it will realize a loss for the protocol.

Similarly, the function `instantRedeem()` does not take into account Midas' fee into consideration. However, it charges its own fee which may be enough to cover for Midas' fee.

Vulnerability Detail

The function `VaultsSolverMidas.deposit()` strictly converts the deposit amount to USD and to mToken to send that to the receiver. It does not take into account any fee:

```
1     function deposit(address _vaultToken, address _depositAsset, address
2         _receiver, uint256 _amount) external {
3         if (!supportedDepositAssetsPerVault[_vaultToken][_depositAsset]) {
4             revert VaultInstantWithdrawalMidas__DepositAssetNotSupported(
5                 _vaultToken, _depositAsset);
6         }
7         (uint256 usdAmount, address depositVault) = _convertTokenToUSD(
8             _vaultToken, _depositAsset, _amount);
9         uint256 mTokenAmount = _convertUSDToMToken(depositVault, usdAmount);
10        _depositAsset.safeTransferFrom(msg.sender, address(this), _amount);
11        _vaultToken.safeTransfer(_receiver, mTokenAmount);
12        emit Deposit(_vaultToken, _depositAsset, _receiver, _amount,
13            mTokenAmount);
14    }
```

On the contrary, the Midas function `depositInstant()` charges a fee to the user:

```
1     function depositInstant(
2         address tokenIn,
3         uint256 amountToken,
4         uint256 minReceiveAmount,
5         bytes32 referrerId,
6         address recipient
7     )
8     external
9     whenFnNotPaused(_DEPOSIT_INSTANT_WITH_CUSTOM_RECIPIENT_SELECTOR)
10    {
11        ...
12        CalcAndValidateDepositResult memory result = _depositInstant(
13            tokenIn,
```

```
14         amountToken,  
15         minReceiveAmount,  
16         recipient  
17     );  
18     ...  
19 }  
20  
21 function _depositInstant(  
22     address tokenIn,  
23     uint256 amountToken,  
24     uint256 minReceiveAmount,  
25     address recipient  
26 ) internal virtual returns (CalcAndValidateDepositResult memory result) {  
27     address user = msg.sender;  
28  
29     result = _calcAndValidateDeposit(user, tokenIn, amountToken, true);  
30  
31     ...  
32     _instantTransferTokensToTokensReceiver(  
33         tokenIn,  
34         result.amountTokenWithoutFee,  
35         result.tokenDecimals  
36     );  
37  
38     if (result.feeTokenAmount > 0)  
39         _tokenTransferFromUser(  
40             tokenIn,  
41             feeReceiver,  
42             result.feeTokenAmount,  
43             result.tokenDecimals  
44         );  
45  
46     mToken.mint(recipient, result.mintAmount);  
47  
48     ...  
49 }
```

The calculation of the output mToken is similar to what is done in `VaultsSolverMidas` with addition of a fee:

```
1     function _calcAndValidateDeposit(  
2         address user,  
3         address tokenIn,  
4         uint256 amountToken,  
5         bool isInstant  
6     ) internal returns (CalcAndValidateDepositResult memory result) {  
7         ...  
8         (uint256 amountInUsd, uint256 tokenInUSDRate) = _convertTokenToUsd(  
9             tokenIn,  
10            amountToken  
11        );  
12        result.tokenAmountInUsd = amountInUsd;  
13        result.tokenInRate = tokenInUSDRate;  
14        address userCopy = user;  
15        ...  
16    }
```

```

16
17     result.feeTokenAmount = _truncate(
18         _getFeeAmount(userCopy, tokenIn, amountToken, isInstant, 0),
19         result.tokenDecimals
20     );
21     result.amountTokenWithoutFee = amountToken - result.feeTokenAmount;
22
23     uint256 feeInUsd = (result.feeTokenAmount * result.tokenInRate) /
24         10**18;
25
26     (uint256 mTokenAmount, uint256 mTokenRate) = _convertUsdToMToken(
27         result.tokenAmountInUsd - feeInUsd
28     );
29     result.mintAmount = mTokenAmount;
30     result.tokenOutRate = mTokenRate;
31     ...
32 }

```

`instantRedeem()` charges a fee but it does not take into account Midas' fee calculation when computing the amount of withdrawal assets to send out:

```

1     function instantRedeem(
2         address _recipient,
3         address _vaultToken,
4         address _withdrawalAsset,
5         uint256 _amount
6     )
7     external
8     {
9         if (!supportedWithdrawalAssetsPerVault[_vaultToken][_withdrawalAsset])
10            {
11                revert VaultInstantWithdrawalMidas__WithdrawalAssetNotSupported(
12                    _vaultToken, _withdrawalAsset);
13            }
14            (uint256 usdAmount, address redemptionVault) = _convertMTokenToUSD(
15                _vaultToken, _amount); // @audit a midas fee is charged before that
16                in RedemptionVault
17            uint256 tokenAmount = _convertUSDToToken(redemptionVault,
18                _withdrawalAsset, usdAmount);
19            uint256 instantRedeemFee = _calcInstantRedeemFee(_vaultToken,
20                tokenAmount);
21            uint256 adminFee = _calcAdminFee(instantRedeemFee);
22            if (adminFee > 0 && adminFeeRecipient != address(0)) {
23                _withdrawalAsset.safeTransfer(adminFeeRecipient, adminFee);
24            }
25            uint256 amountToTransfer = _amount - instantRedeemFee;
26            _vaultToken.safeTransferFrom(msg.sender, address(this), _amount);
27            _withdrawalAsset.safeTransfer(_recipient, tokenAmount);
28            emit InstantRedeem(_vaultToken, _withdrawalAsset, _amount,
29                amountToTransfer, instantRedeemFee, adminFee);
30        }
31    }

```

On Midas' `RedemptionVault` the first step of an instant redeem is to call `_calcAndValidateRedeem()`


```
1     function _redeemInstant(  
2         address tokenOut,  
3         uint256 amountMTokenIn,  
4         uint256 minReceiveAmount,  
5         address recipient  
6     )  
7     internal  
8     virtual  
9     returns (  
10         CalcAndValidateRedeemResult memory calcResult,  
11         uint256 amountTokenOutWithoutFee  
12     )  
13     {  
14         address user = msg.sender;  
15  
16         calcResult = _calcAndValidateRedeem(  
17             user,  
18             tokenOut,  
19             amountMTokenIn,  
20             true,  
21             false  
22         );  
23         ...  
24     }
```

The `_calcAndValidateRedeem()` function charges a fee diminishing the amount of mToken later converted into output asset:

```
1     function _calcAndValidateRedeem(  
2         address user,  
3         address tokenOut,  
4         uint256 amountMTokenIn,  
5         bool isInstant,  
6         bool isFiat  
7     ) internal view returns (CalcAndValidateRedeemResult memory result) {  
8         require(amountMTokenIn > 0, "RV: invalid amount");  
9  
10        if (!isFreeFromMinAmount[user]) {  
11            uint256 minRedeemAmount = isFiat ? minFiatRedeemAmount : minAmount;  
12            require(minRedeemAmount <= amountMTokenIn, "RV: amount < min");  
13        }  
14  
15        result.feeAmount = _getFeeAmount(  
16            user,  
17            tokenOut,  
18            amountMTokenIn,  
19            isInstant,  
20            isFiat ? fiatAdditionalFee : 0  
21        );  
22  
23        ...  
24  
25        result.amountMTokenWithoutFee = amountMTokenIn - result.feeAmount;  
26    }
```

Impact

Users can exchange assets for mTokens at a better rate on `VaultsSolverMidas` than on Midas' contracts. If the owner of the solver wants to exchange mTokens back to assets, they will have to pay a fee realizing an overall loss for the protocol.

The situation may be the same for redeeming mTokens into other assets.

This may be intended behaviour for the protocol as a means to balance out the supply of mTokens / other assets on the contract.

Code Snippets

<https://github.com/midas-apps/contracts/blob/ba8596751be53f4461349316b380c1e7868782e5/contracts/DepositVault.sol>

<https://github.com/midas-apps/contracts/blob/ba8596751be53f4461349316b380c1e7868782e5/contracts/RedemptionVault.sol>

<https://github.com/0xhyperbeat/DNCoreWriter/blob/b0e409d6e53ebf17cf881f3a323acdcd60c47b85/src/VaultsSolverMidas.sol#L87-L126>

Recommendation

Acknowledge that this is intended behaviour.

Alternatively, calculate the fee Midas would take on a deposit / withdraw and charge that same fee in `deposit()` and `instantRedeem()`.

Response

The protocol team acknowledged that this is intended behaviour.

[Ack][Info] Complete reliance on Midas oracles

Summary

The solver completely rely on oracles controlled by Midas for its conversion. If Midas decides to siphon all the funds of the contract, they can do that by updating their oracles to devalue the output tokens or overvalue the input tokens.

Vulnerability Detail

The function to convert the mToken amount to USD value queries the oracle of underlying Midas vault:

```
1     function _convertMTokenToUSD(  
2         address _vaultToken,  
3         uint256 _amount  
4     )  
5         internal  
6         view  
7         returns (uint256 usdAmount, address redemptionVault)  
8     {  
9         redemptionVault = redemptionVaults[_vaultToken];  
10        if (redemptionVault == address(0)) {  
11            revert VaultInstantWithdrawalMidas__RedemptionVaultNotSet(  
12                _vaultToken);  
13        }  
14        address mTokenDataFeed = IMidasRedemptionVault(redemptionVault).  
15            mTokenDataFeed();  
16        usdAmount = IMidasDataFeed(mTokenDataFeed).getDataInBase18().mulDiv(  
17            _amount, 10 ** 18);  
18    }
```

The Midas oracle gets the answer by querying an aggregator round data:

```
1     function getDataInBase18() external view returns (uint256 answer) {  
2         (, answer) = _getDataInBase18();  
3     }  
4  
5     function _getDataInBase18()  
6         private  
7         view  
8         returns (uint80 roundId, uint256 answer)  
9     {  
10        uint8 decimals = aggregator.decimals();  
11        (uint80 _roundId, int256 _answer, , uint256 updatedAt, ) = aggregator  
12            .latestRoundData();  
13        require(_answer > 0, "DF: feed is deprecated");  
14        require(  
15            // solhint-disable-next-line not-rely-on-time  
16            block.timestamp - updatedAt <= healthyDiff &&
```

```
17         _answer >= minExpectedAnswer &&
18         _answer <= maxExpectedAnswer,
19         "DF: feed is unhealthy"
20     );
21     roundId = _roundId;
22     answer = uint256(_answer).convertToBase18(decimals);
23 }
```

The used aggregator can be updated at any time by Midas admin:

```
1     function changeAggregator(address _aggregator)
2         external
3         onlyRole(feedAdminRole(), msg.sender)
4     {
5         require(_aggregator != address(0), "DF: invalid address");
6
7         aggregator = AggregatorV3Interface(_aggregator);
8     }
```

Impact

Midas admins can set up a forged aggregator at any point in time to control the exchange rate used by [VaultsSolverMidas](#) and profit from that to siphon all the funds held by the contract.

Code Snippets

<https://github.com/0xhyperbeat/DNCoreWriter/blob/b0e409d6e53ebf17cf881f3a323acdc60c47b85/src/VaultsSolverMidas.sol#L173>

<https://github.com/midas-apps/contracts/blob/ba8596751be53f4461349316b380c1e7868782e5/contracts/feeds/DataFeed.sol#L74-L81>

Recommendation

Acknowledge that Midas is trusted with the funds held by [VaultsSolverMidas](#).

Response

The underlying Midas vaults will be owned by the protocol themselves so relying on them is not an issue.

Optimisations and miscellaneous

This part lists minor gas/code optimizations that shouldn't make the code less readable or improve overall readability. It also lists questions about unclear code segments.

Order of arguments in redeem and deposit do not match

The arguments used in the functions `instantRedeem()` and `deposit()` are similar but their orders do not match which may lead to errors when the functions are called.

`instantRedeem()` uses `(recipient, vaultToken, withdrawalAsset, amount)` while `deposit()` uses `(vaultToken, depositAsset, receiver, amount)`.

<https://github.com/0xhyperbeat/DNCoreWriter/blob/b0e409d6e53ebf17cf881f3a323acdc60c47b85/src/VaultsSolverMidas.sol#L117>